```
Reduce (CSL, rev 6657), 10-Dec-2023 ...

1: in "D:\OneDrive\work\ConformalFieldTheory\REDUCE\virasoro_algebra.red";
on rat;


on div;



% [FF] B. L. Feigin & D. B. Fuchs, Verma modules over the virasoro algebra
% Lecture Notes in Mathematics, Volume 1060, 1984, pp. 230--245,
% Topology,
% General and Algebraic Topology and Applications,
% Proceedings of the International Topological Conference held in Leningrad, August 23-27, 1983

% delta - Knonecker's delta (delta(m) = if m = 0 then 1 else 0)
% c - central charge
% h - conformal weight
% x - unknowns
% z - unknowns
% tau - Use t instead of t because t is the symbol for true in REDUCE
% theta - theta^2 = tau
% l - generators of the Virasoro algebra
% vec - ket |h> or bra <h| vector
% vecproj - L(m)*vevproj() = 0 if m < -2
% gamma - mu + 2h of [FF]
% vecf - w^{-gamma}

clear delta, c, h, x, z, tau, theta, l, vec, vecproj, gamma, vecf;


order h, c, x, z, theta, tau, l;


operator delta, l, vec, vecproj;


noncom l, vec, vec_proj, vecf;


factor l, vec, gamma, vecf;



let theta**2 = tau;


for all m let delta(m) = if m = 0 then 1 else 0;


for all m, n such that m > n let l(m)*l(n) = l(n)*l(m) + (m-n)*l(m+n) + c*(m**3-m)*delta(m+n)/12;


for all h let vec(h)**2 = 1;


for all m, n, h such that m > 0 let l(m)*vec(h) = 0;


for all m, n, h such that 0 > n let vec(h)*l(n) = 0;


for all h let l(0)*vec(h) = h*vec(h), vec(h)*l(0) = vec(h)*h;


let vecproj()**2 = 1;


for all m let vecproj()*l(m) = if m < -2 then 0 else if m = -1 then z1*vecproj() else z2*vecproj();


for all h, gamma, epsilon let vecf(h, gamma)*vecf(h, epsilon) = 1;


*** vecf declared operator

for all m, h, gamma let l(m)*vecf(h, gamma) = -(-gamma + h*(m+1)) * vecf(h, gamma - m);


procedure commutator(x, y); x*y - y*x;
```

$commutator$

```
procedure centralcharge(tau); 6/tau + 13 + 6*tau;
```

centralcharge

```
procedure conformalweight(r, s, tau) = (1-r**2)/(4*tau) + (1-r*s)/2 + (1-s**2)/4*tau;
```

conformalweight

```
procedure cw(r, s, tau); conformalweight(r, s, tau);
```

cw

```
procedure partitions(n); begin
  if n < 0 then return {};
  if n = 0 then return {{}};
  if n = 1 then return {{1}};
  return append(
    for each p in partitions(n-1) collect 1 . p,
    for each p in partitions(n-1) join
      if length(p)=1 or first(p) < second(p) then {(first(p)+1) . rest(p)} else {}
  );
end;
```

partitions

```
% partnum(n) = partition number of n
procedure partnum(n); length(partitions(n));
```

partnum

```
% Define degree by deg(L(-m)) = m
% monomial({a,b,c,...}) = ...*L(-c)*L(-b)*L(-a) (a,b,c,...>0)
% monomials(d) = list of all monomials of degree d
% dualmonomial({a,b,c,...}) = l(a)*L(b)*L(c)*...
% dualmonomials(d) = list of all dualmonomials of degree -d

procedure monomial(p); for each m in p product l(-m);
```

monomial

```
procedure monomials(d); for each p in partitions(d) collect monomial(p);
```

monomials

```
procedure dualmonomial(p); for each m in reverse(p) product l(m);
```

dualmonomial

```
procedure dualmonomials(d); for each p in partitions(d) collect dualmonomial(p);
```

dualmonomials

```
% kacmat(d) = p(d)xp(d) matrix determinant of which is equal to Kac determinant of degree d
procedure kacmat(d); begin
  scalar basis_r := monomials(d);
  scalar basis_l := dualmonomials(d);
  scalar N := partnum(d);
  scalar j := 0;
  scalar k := 0;
  matrix kac_matrix(N, N);
  for each ll_r in basis_r do <<
    k := k + 1;
    for each ll_l in basis_l do <<
      j := j + 1;
      kac_matrix(j, k) := vec(h) * ll_l * ll_r * vec(h);
    >>;
    j := 0;
  >>;
  return kac_matrix;
end;
```

kacmat

```
% kacdet(d) = Kac determinant of degree d
procedure kacdet(d); begin
  kac_matrix := kacmat(d);
  return det(kac_matrix);
end;
```

kacdet

```
% kacdet_t(d, tau) = Kac determinant of degree d for c = c(tau)
procedure kacdet_t(d, tau); sub(c = centralcharge(tau), kacdet(d));
```

$kacdet_t$

```
% kacdet_t_rhs(d, tau) = formula of kacdet_t(d, tau)
procedure kacdet_t_rhs(d, tau); begin
  return for r := 1:d product for s := 1:d product (h - cw(r, s, tau))^partnum(d - r*s);
end;
```

kacdet_t_rhs

```
% kacdet_fact(d, tau) = factorization of kacdet_t(d, tau)
procedure kac
det_fact(d, tau); begin
  scalar kd := kacdet_t(d, tau);
  return {num = factorize(num(kd)), den = den(kd)};
end;
```

kacdet_fact

```
% kacdet_sol(d, tau) = solution of kacdet_t(d, tau) w.r.t. h
procedure kacdet_sol(d, tau); begin
  scalar kd := kacdet_t(d, tau);
  return solve(kd, h)
end;
```

kacdet_sol

```
% lincomb(d) = L(-1)^d + x1 L(-2)*L(-1)^{d-2} + ...
procedure lincomb(d); begin
  scalar j := -1;
  return for each ll in monomials(d) sum <<
    j := j + 1;
    if j = 0 then ll else mkid(x, j) * ll
  >>;
end;
```

lincomb

```
% ltoz(s, d) = replace l(-m) in s by z_m
procedure ltoz(s, d); begin
  scalar ss := s;
  for m := 1:d do ss := (ss where l(-m) => mkid(z, m));
  return ss;
end;
```

ltoz

```
% allcoeffs(s, d) = all coefficients of s
procedure allcoeffs(s, d); begin
  scalar sz := ltoz(s, d);
  scalar cs := coeff(sz, mkid(z, 1));
  for i := 2:d do
    (cs := for each f in cs join for each g in coeff(f, mkid(z, i)) join if g = 0 then {} else {g});
  return cs;
end;
```

allcoeffs

```
% solsingvec(r, s, tau) = solution x1, x2, ... of the equation of the singular vector for (r, s)
procedure solsingvec(r, s, tau); begin
  scalar d, sing, v1, s1, lineq1, v2, s2, lineq2, lineq, lineq_t, xs, sol;
  if r = 1 and s = 1 then return {{}};
  d := r*s;
  sing := lincomb(d);
  v1 := l(1)*sing*vec(h);
  s1 := ltoz(v1*vec(h), d);
  lineq1 := allcoeffs(s1, d);
  v2 := l(2)*sing*vec(h);
  s2 := ltoz(v2*vec(h), d);
  lineq2 := allcoeffs(s2, d);
  lineq := append(lineq1, lineq2);
  lineq_t := sub(c = centralcharge(tau), h = conformalweight(r, s, tau), lineq);
  xs := for i := 1:(length(partitions(d)) - 1) collect mkid(x, i);
  sol := solve(lineq_t, xs);
  return sol;
end;
```

solsingvec

```
% singvec(r, s, tau) = Virasoro part of the singular vector for (r, s)
procedure singvec(r, s, tau); begin
  scalar sing := lincomb(r*s);
  scalar sol := solsingvec(r, s, tau);
  return sub(first(sol), sing)
end;
```

singvec

```
% proj12(s) = replace l(m) (m<-2) in s by 0 and l(-1), l(-2) by z1, z2 respectively
procedure proj12(s); begin
  scalar sp := vecproj() * s;
  return vecproj() * sp;
end;
```

$proj_{12}$

```
% proj12_fact(s) = factorization of proj12(s)
procedure proj12_fact(s); begin
  scalar sp := vecproj() * s;
  sp := vecproj() * sp;
  return {num = factorize(num(sp)), den = den(sp)};
end;
```

proj12_fact

```
% act_ff(s, h, gamma) = result of the action of s on w^{-gamma} with conformal weight h
procedure act_ff(s, h, gamma); begin
  scalar g := s * vecf(h, gamma);
  g := g * vecf(h, gamma);
  return g;
end;
```

act_ff

```
% g_ff(r, s, h, gamma, tau) = result of the action of singvec(r, s, tau) on w^{-gamma} with h
procedure g_ff(r, s, h, gamma, tau); begin
  scalar sing := singvec(r, s, tau);
  return act_ff(sing, h, gamma);
end;
```

g_ff

```
% g_ff_factor(k, l, a, b, h, gamma) = the formula of Theorem 3.2 in [FF]
procedure g_ff_factor(k, l, a, b, h, gamma); begin
  return (gamma^2
    + ((2*a*(k-a)+k)/theta^2 + (2*b*(l-b)+l)*theta^2 + k*l+k+l-(k-2*a)*(l-2*b)) * gamma
    + ((k-2*a)/theta + (l-2*b)*theta)^2 * h
    + (a/theta + b*theta) * ((a+1)/theta + (b+1)*theta) *
```

```
      ((k-a)/theta + (l-b)*theta) * ((k-a+1)/theta + (l-b+1)*theta)
  );
end;
```

g_ff_factor

```
% g_ff_factor_lhs(r0, s0, r1, s1, h2, i1, j1)
% = another version of the formula in Theorem 3.2 of [FF]
procedure g_ff_factor_lhs(r0, s0, r1, s1, h2, i1, j1); begin
  scalar k := r1 - 1;
  scalar l := s1 - 1;
  scalar a := i1 - 1;
  scalar b := j1 - 1;
  scalar h := cw(r0, s0, tau);
  scalar gamma := cw(r0, s0, tau) + cw(r1, s1, tau) - h2;
  return g_ff_factor(k, l, a, b, h, gamma);
end;
```

g_ff_factor_lhs

```
% g_ff_factor_rhs(r0, s0, r1, s1, h2, i1, j1)
% = explicit factorization of g_ff_factor_lhs(r0, s0, r1, s1, h2, i1, j1)
procedure g_ff_factor_rhs(r0, s0, r1, s1, h2, i1, j1); begin
  return (
    (h2 - cw(r0+r1+1-2*i1, s0+s1+1-2*j1, tau)) *
    (h2 - cw(r0-(r1+1-2*i1), s0-(s1+1-2*j1), tau))
  );
end;
```

g_ff_factor_rhs

```
% f_ff(r1, s1, h0, h1, h2, tau)
% = <h2|Phi(w, singvec_{r1
,s1}|h1>)|h0> / w^{h2-h1-h0-r1*s1}
% where c = c(tau)
procedure f_ff(r1, s1, h0, h1, h2, tau); g_ff(r1, s1, h0, h0 + h1 - h2, tau);
```

f_ff

```
% f_fusion(r0, s0, r1, s1, h2, tau)
% = <h2|Phi(w, singvec_{r1,s1}|h1>)|h0> / w^{h2-h1-h0-r1*s1}
% where c = c(tau), h0 = h_{r0,s0}(tau), h1 = h_{r1,s1}(tau).
procedure f_fusion(r0, s0, r1, s1, h2, tau); begin
  scalar h0 := cw(r0, s0, tau);
  scalar h1 := cw(r1, s1, tau);
  return f_ff(r1, s1, h0, h1, h2, tau);
end;
```

f_fusion

```
% f_fusion_rhs(r0, s0, r1, s1, h2, tau)
% = the explicit factorization of f_fusion(r0, s0, r1, s1, h2, tau)
procedure f_fusion_rhs(r0, s0, r1, s1, h2, tau); begin
  return ((-1)^(r1*s1) *
    (for i1 := 1:r1 product for j1 := 1:s1 product h2 - cw(r0+r1+1-2*i1, s0+s1+1-2*j1, tau))
  );
end;
```

f_fusion_rhs

```
% f_fusion_fact(r0, s0, r1, s1, h2, tau)
% = factorization of f_ff(r1, s1, h0, h1, h2, tau)
procedure f_fusion_fact(r0, s0, r1, s1, h2, tau); begin
  scalar f := f_fusion(r0, s0, r1, s1, h2, tau);
  return {num = factorize(num(f)), den = den(f)};
end;
```

f_fusion_fact

```
% f_fusion_fact(r0, s0, r1, s1, h2, tau)
% = solution of f_ff(r1, s1, h0, h1, h2, tau) = 0 w.r.t. h2
```

```
procedure f_fusion_sol(r0, s0, r1, s1, h2, tau); begin
  scalar f := f_fusion(r0, s0, r1, s1, h2, tau);
  return solve(f, h2);
end;
```

$$f\_fusion\_sol$$

```
% f_minimal_model(p, q) = <h|Phi(w, singvec_{p-1,q-1}|0>)|h> / w^{-r1*s1}
% where c = c(-p/q)
procedure f_minimal_model(p, q); f_ff(p-1, q-1, h, 0, h, -p/q);
```

$$f\_minimal\_model$$

```
% f_minimal_model_fact(p, q) = factorization of f_minimal_model(p, q)
procedure f_minimal_model_fact(p, q); begin
  scalar f := f_minimal_model(p, q);
  scalar topcoeff := first(reverse(coeff(f, h)));
  scalar sol := solve(f, h);
  return {topcoeff, sol};
end;
```

$$f\_minimal\_model\_fact$$

```
% f_minimal_model_factor_lhs(p, q)
% = certain specialization of the formula of Theorem 3.2 in [FF].
procedure f_minimal_model_factor_lhs(p, q, r, s, h); begin
  return sub(tau=-p/q, g_ff_factor(p-2, q-2, r-1, s-1, h, 0));
end;
```

$$f\_minimal\_model\_factor\_lhs$$

```
% f_minimal_model_factor_rhs(p, q)
% = explicit simplification of f_minimal_model_factor_lhs(p, q)
procedure f_minimal_model_factor_rhs(p, q, r, w, h); begin
  return -4*(q*r - p*s)^2/(p*q) * (h - cw(r, s, -p/q));
end;
```

$$f\_minimal\_model\_factor\_rhs$$

```
% table_minimal_model(p, q)
% = table of the conformal weights of the minimal model for c=c(-p/q).
procedure table_minimal_model(p, q); begin
  matrix table_mm(p-1, q-1);
  for r := 1:(p-1) do for s := 1:(q-1) do begin
    table_mm(r, s) := cw(r, s, -p/q);
  end;
  return table_mm;
end;
```

$$table\_minimal\_model$$

```
;

end;

2: in "D:\OneDrive\work\ConformalFieldTheory\REDUCE\virasoro_algebra_test.red";
% L(m)'s satidfy the relation of the Virasoro algebta.

operator O;


noncom O;


commutator(O(m), O(n));
```

$$o(m)\ o(n) - o(n)\ o(m)$$

```
commutator(L(0), L(-2));
```

$$2\,l\,(-2)$$

```
commutator(L(0), L(-1));
```

$$l\,(-1)$$

```
commutator(L(0), L(0));
```

$$0$$

```
commutator(L(0), L(1));
```

$$-l\,(1)$$

```
commutator(L(0), L(2));
```

$$-2\,l\,(2)$$

```
commutator(L(1), L(-2));
```

$$3\,l\,(-1)$$

```
commutator(L(1), L(-1));
```

$$2\,l\,(0)$$

```
commutator(L(1), L(0));
```

$$l\,(1)$$

```
commutator(L(1), L(1));
```

$$0$$

```
commutator(L(1), L(2));
```

$$-l\,(3)$$

```
commutator(L(2), L(-2));
```

$$4\,l\,(0)+\frac{1}{2}\,c$$

```
commutator(L(2), L(-1));
```

$$3\,l\,(1)$$

```
commutator(L(2), L(0));
```

$$2\,l\,(2)$$

```
commutator(L(2), L(1));
```

$$l\,(3)$$

```
commutator(L(2), L(2));
```

$$0$$

```
% left action of L(m) on |h>
L(-2)*vec(h);
```

$$l\,(-2)\ vec\,(h)$$

```
L(-1)*vec(h);
```

$$l(-1)\;vec(h)$$

```
L(0)*vec(h);
```

$$h\,vec(h)$$

```
L(1)*vec(h);
```

$$0$$

```
L(2)*vec(h);
```

$$0$$

```
% right action of L(m) on <h|
vec(h)*L(-2);
```

$$0$$

```
vec(h)*L(-1);
```

$$0$$

```
vec(h)*L(0);
```

$$h\,vec(h)$$

```
vec(h)*L(1);
```

$$vec(h)\;l(1)$$

```
vec(h)*L(2);
```

$$vec(h)\;l(2)$$

```
% <h|h> = 1
vec(h)*vec(h);
```

$$1$$

```
% central charge c(tau)
centralcharge(tau);
```

$$\tau^{-1}\,\left(6\,\tau^2+13\,\tau+6\right)$$

```
% conformal weight h_{r,s}(tau)
conformalweight(r, s, tau);
```

$$\tau^{-1}\,\left(-\frac{1}{4}\,\tau^2\,s^2+\frac{1}{4}\,\tau^2-\frac{1}{2}\,\tau\,r\,s+\frac{1}{2}\,\tau-\frac{1}{4}\,r^2+\frac{1}{4}\right)$$

```
cw(r, s, tau);
```

$$\tau^{-1}\,\left(-\frac{1}{4}\,\tau^2\,s^2+\frac{1}{4}\,\tau^2-\frac{1}{2}\,\tau\,r\,s+\frac{1}{2}\,\tau-\frac{1}{4}\,r^2+\frac{1}{4}\right)$$

```
% list of the all partition of n
partitions(5);
```

$$\{\{1,1,1,1,1\},\{1,1,1,2\},\{1,1,3\},\{1,2,2\},\{1,4\},\{2,3\},\{5\}\}$$

```
% partition number of n
partnum(5);
```

$$7$$

```
% monomials and dual monomials of L(m)
monomial({1,1,2,3,3,3,4});
```

$$l(-4)\ l(-3)^3\ l(-2)\ l(-1)^2$$

```
monomials(5);
```

$$\left\{l(-1)^5, l(-2)\ l(-1)^3, l(-3)\ l(-1)^2, l(-2)^2\ l(-1), l(-4)\ l(-1), l(-3)\ l(-2), l(-5)\right\}$$

```
dualmonomial({1,1,2,3,3,3,4});
```

$$l(1)^2\ l(2)\ l(3)^3\ l(4)$$

```
dualmonomials(5);
```

$$\left\{l(1)^5, l(1)^3\ l(2), l(1)^2\ l(3), l(1)\ l(2)^2, l(1)\ l(4), l(2)\ l(3), l(5)\right\}$$

```
% Kac determinant
kacmat(2);
```

$$\begin{pmatrix} 4h\ (2h+1) & 6h \\ 6h & 4h+\frac{1}{2}c \end{pmatrix}$$

```
kacdet(2);
```

$$2h\ \left(16h^2+2hc-10h+c\right)$$

```
kacdet_t(2, tau);
```

$$2\tau^{-1}h\ \left(16h^2\tau+12h\tau^2+16h\tau+12h+6\tau^2+13\tau+6\right)$$

```
kacdet_fact(2, tau);
```

$$\{num=\{\{2,1\},\{4h\tau+2\tau+3,1\},\{4h+3\tau+2,1\},\{h,1\}\}, den=\tau\}$$

```
kacdet_sol(2, tau);
```

$$\left\{h=\frac{-2\tau-3}{4\tau}, h=\frac{-3\tau-2}{4}, h=0\right\}$$

```
kd_rhs := kacdet_t_rhs(2, tau);
```

$$kd\_rhs := \tau^{-1}h\ \left(h^2\tau+\frac{3}{4}h\tau^2+h\tau+\frac{3}{4}h+\frac{3}{8}\tau^2+\frac{13}{16}\tau+\frac{3}{8}\right)$$

```
kacdet_t(2, tau) / kd_rhs;
```

$$32$$

```
{num = factorize(num(kd_rhs)), den = den(kd_rhs)};
```

$$\{num=\{\{4h\tau+2\tau+3,1\},\{4h+3\tau+2,1\},\{h,1\}\}, den=16\tau\}$$

```
solve(kd_rhs, h);
```

$$\left\{h=\tau^{-1}\left(-\frac{1}{2}\tau-\frac{3}{4}\right), h=-\frac{3}{4}\tau-\frac{1}{2}, h=0\right\}$$

```
kacmat(3);
```

$$mat\left(\left(24\,h\,\left(2\,h^2+3\,h+1\right),12\,h\,\left(3\,h+1\right),24\,h\right),\left(12\,h\,\left(3\,h+1\right),h\,\left(8\,h+c+8\right),10\,h\right),\left(24\,h,10\,h,2\,\left(3\,h+c\right)\right)\right)$$

```
kacdet(3);
```

$$48\,h^2\,\left(48\,h^4+22\,h^3\,c-142\,h^3+2\,h^2\,c^2-5\,h^2\,c+102\,h^2+3\,h\,c^2-13\,h\,c-20\,h+c^2+2\,c\right)$$

```
kacdet_t(3, tau);
```

$$144\,\tau^{-2}\,h^2\times\left(16\,h^4\,\tau^2+44\,h^3\,\tau^3+48\,h^3\,\tau^2+44\,h^3\,\tau+24\,h^2\,\tau^4+94\,h^2\,\tau^3+173\,h^2\,\tau^2+94\,h^2\,\tau+24\,h^2+36\,h\,\tau^4+130\,h\,\tau^3+178\,h\,\tau^2+130\,h\,\tau+36\,h+12\,\tau^4+56\,\tau^3+89\,\tau^2+56\,\tau+12\right)$$

```
kacdet_fact(3, tau);
```

$$\left\{num=\left\{\{144,1\},\{4\,h\,\tau+2\,\tau+3,1\},\{h\,\tau+\tau+2,1\},\{4\,h+3\,\tau+2,1\},\{h+2\,\tau+1,1\},\{h,2\}\right\},den=\tau^2\right\}$$

```
kacdet_sol(3, tau);
```

$$\left\{h=0,h=\frac{-\tau-2}{\tau},h=-2\,\tau-1,h=\frac{-2\,\tau-3}{4\,\tau},h=\frac{-3\,\tau-2}{4}\right\}$$

```
kd_rhs := kacdet_t_rhs(3, tau);
```

$$kd\_rhs:=\tau^{-2}\,h^2\times\left(h^4\,\tau^2+\frac{11}{4}\,h^3\,\tau^3+3\,h^3\,\tau^2+\frac{11}{4}\,h^3\,\tau+\frac{3}{2}\,h^2\,\tau^4+\frac{47}{8}\,h^2\,\tau^3+\frac{173}{16}\,h^2\,\tau^2+\frac{47}{8}\,h^2\,\tau+\frac{3}{2}\,h^2+\frac{9}{4}\,h\,\tau^4+\frac{65}{8}\,h\,\tau^3+\frac{89}{8}\,h\,\tau^2+\frac{65}{8}\,h\,\tau+\frac{9}{4}\,h+\frac{3}{4}\,\tau^4+\frac{7}{2}\,\tau^3+\frac{89}{16}\,\tau^2+\frac{7}{2}\,\tau+\frac{3}{4}\right)$$

```
kacdet_t(3, tau) / kd_rhs;
```

$$2304$$

```
{num = factorize(num(kd_rhs)), den = den(kd_rhs)};
```

$$\left\{num=\left\{\{4\,h\,\tau+2\,\tau+3,1\},\{h\,\tau+\tau+2,1\},\{4\,h+3\,\tau+2,1\},\{h+2\,\tau+1,1\},\{h,2\}\right\},den=16\,\tau^2\right\}$$

```
solve(kd_rhs, h);
```

$$\left\{h=0,h=\tau^{-1}\,\left(-\tau-2\right),h=-2\,\tau-1,h=\tau^{-1}\,\left(-\frac{1}{2}\,\tau-\frac{3}{4}\right),h=-\frac{3}{4}\,\tau-\frac{1}{2}\right\}$$

```
kacmat(4);
```

$$mat\left(\left(96\,h\,\left(4\,h^3+12\,h^2+11\,h+3\right),48\,h\,\left(6\,h^2+7\,h+2\right),48\,h\,\left(4\,h+1\right),72\,h\,\left(3\,h+2\right),120\,h\right),\left(48\,h\,\left(6\,h^2+7\,h+2\right),2\,h\,\left(16\,h^2+2\,h\,c+58\,h+c+16\right),20\,h\,\left(2\,h+1\right),6\,h\,\left(8\,h+c+8\right),36\,h\right),\left(48\,h\,\left(4\,h+1\right),20\,h\,\left(2\,h+1\right),4\,h\,\left(3\,h+c+3\right),30\,h,14\,h\right),\left(72\,h\,\left(3\,h+2\right),6\,h\,\left(8\,h+c+8\right),30\,h,32\,h^2+8\,h\,c+32\,h+\frac{1}{2}\,c^2+4\,c,3\,\left(8\,h+c\right)\right),\left(120\,h,36\,h,14\,h,3\,\left(8\,h+c\right),8\,h+5\,c\right)\right)$$

```
kacdet(4);
```

$$384\,h^3\times$$

$$
\begin{gathered}
\left(98304\,h^{9}+131072\,h^{8}\,c-868352\,h^{8}+60416\,h^{7}\,c^{2}-612352\,h^{7}\,c+2708480\,h^{7}+12288\,h^{6}\,c^{3}-\right.\\
100864\,h^{6}\,c^{2}+876288\,h^{6}\,c-3951872\,h^{6}+1144\,h^{5}\,c^{4}+5664\,h^{5}\,c^{3}+1584\,h^{5}\,c^{2}-446880\,h^{5}\,c+\\
2969432\,h^{5}+40\,h^{4}\,c^{5}+2420\,h^{4}\,c^{4}-10620\,h^{4}\,c^{3}+71052\,h^{4}\,c^{2}-41500\,h^{4}\,c-1153232\,h^{4}+140\,h^{3}\,c^{5}+\\
1058\,h^{3}\,c^{4}+1140\,h^{3}\,c^{3}-37630\,h^{3}\,c^{2}+107116\,h^{3}\,c+210440\,h^{3}+170\,h^{2}\,c^{5}-221\,h^{2}\,c^{4}+2175\,h^{2}\,c^{3}+\\
3340\,h^{2}\,c^{2}-28984\,h^{2}\,c-13200\,h^{2}+85\,h\,c^{5}+18\,h\,c^{4}-1233\,h\,c^{3}+434\,h\,c^{2}+2640\,h\,c+15\,c^{5}+81\,c^{4}+36\,c^{3}-\\
\left.132\,c^{2}\right)
\end{gathered}
$$

```
kacdet_t(4, tau);
```

$$
\begin{gathered}
2304\,\tau^{-5}\,h^{3}\times\\
\left(16384\,h^{9}\,\tau^{5}+131072\,h^{8}\,\tau^{6}+139264\,h^{8}\,\tau^{5}+131072\,h^{8}\,\tau^{4}+362496\,h^{7}\,\tau^{7}+958464\,h^{7}\,\tau^{6}+\right.\\
1551360\,h^{7}\,\tau^{5}+958464\,h^{7}\,\tau^{4}+362496\,h^{7}\,\tau^{3}+442368\,h^{6}\,\tau^{8}+2270208\,h^{6}\,\tau^{7}+5810944\,h^{6}\,\tau^{6}+\\
7438848\,h^{6}\,\tau^{5}+5810944\,h^{6}\,\tau^{4}+2270208\,h^{6}\,\tau^{3}+442368\,h^{6}\,\tau^{2}+247104\,h^{5}\,\tau^{9}+2345472\,h^{5}\,\tau^{8}+\\
9283392\,h^{5}\,\tau^{7}+19555840\,h^{5}\,\tau^{6}+25163456\,h^{5}\,\tau^{5}+19555840\,h^{5}\,\tau^{4}+9283392\,h^{5}\,\tau^{3}+\\
2345472\,h^{5}\,\tau^{2}+247104\,h^{5}\,\tau+51840\,h^{4}\,\tau^{10}+1084320\,h^{4}\,\tau^{9}+6840720\,h^{4}\,\tau^{8}+22274592\,h^{4}\,\tau^{7}+\\
43663632\,h^{4}\,\tau^{6}+54205920\,h^{4}\,\tau^{5}+43663632\,h^{4}\,\tau^{4}+22274592\,h^{4}\,\tau^{3}+6840720\,h^{4}\,\tau^{2}+\\
1084320\,h^{4}\,\tau+51840\,h^{4}+181440\,h^{3}\,\tau^{10}+2194128\,h^{3}\,\tau^{9}+11446416\,h^{3}\,\tau^{8}+33709164\,h^{3}\,\tau^{7}+\\
62429168\,h^{3}\,\tau^{6}+76354508\,h^{3}\,\tau^{5}+62429168\,h^{3}\,\tau^{4}+33709164\,h^{3}\,\tau^{3}+11446416\,h^{3}\,\tau^{2}+\\
2194128\,h^{3}\,\tau+181440\,h^{3}+220320\,h^{2}\,\tau^{10}+2339064\,h^{2}\,\tau^{9}+11108988\,h^{2}\,\tau^{8}+30950082\,h^{2}\,\tau^{7}+\\
55720647\,h^{2}\,\tau^{6}+67515498\,h^{2}\,\tau^{5}+55720647\,h^{2}\,\tau^{4}+30950082\,h^{2}\,\tau^{3}+11108988\,h^{2}\,\tau^{2}+\\
2339064\,h^{2}\,\tau+220320\,h^{2}+110160\,h\,\tau^{10}+1197288\,h\,\tau^{9}+5711508\,h\,\tau^{8}+15817446\,h\,\tau^{7}+\\
28269126\,h\,\tau^{6}+34152444\,h\,\tau^{5}+28269126\,h\,\tau^{4}+15817446\,h\,\tau^{3}+5711508\,h\,\tau^{2}+1197288\,h\,\tau+\\
110160\,h+19440\,\tau^{10}+228096\,\tau^{9}+1162728\,\tau^{8}+3390120\,\tau^{7}+6259707\,\tau^{6}+7647318\,\tau^{5}+\\
\left.6259707\,\tau^{4}+3390120\,\tau^{3}+1162728\,\tau^{2}+228096\,\tau+19440\right)
\end{gathered}
$$

```
kacdet_fact(4, tau);
```

$$
\begin{gathered}
\{num=\\
\{\{2304,1\},\{4\,h\,\tau+3\,\tau^{2}+6\,\tau+3,1\},\{4\,h\,\tau+6\,\tau+15,1\},\{h\,\tau+\tau+2,1\},\{4\,h+15\,\tau+6,1\},\\
\{h+2\,\tau+1,1\},\{4\,h\,\tau+2\,\tau+3,2\},\{4\,h+3\,\tau+2,2\},\{h,3\}\},den=\tau^{5}\}
\end{gathered}
$$

```
kacdet_sol(4, tau);
```

$$
\left\{h=0,h=\frac{-2\,\tau-3}{4\,\tau},h=\frac{-3\,\tau-2}{4},h=\frac{-3\,\tau^{2}-6\,\tau-3}{4\,\tau},h=\frac{-\tau-2}{\tau},h=-2\,\tau-1,h=\frac{-6\,\tau-15}{4\,\tau},h=\frac{-15\,\tau-6}{4}\right\}
$$

```
kd_rhs := kacdet_t_rhs(4, tau);
```

$$
\begin{gathered}
kd\_rhs:=\\
\tau^{-5}\,h^{3}\times\\
\left(h^{9}\,\tau^{5}+8\,h^{8}\,\tau^{6}+\frac{17}{2}\,h^{8}\,\tau^{5}+8\,h^{8}\,\tau^{4}+\frac{177}{8}\,h^{7}\,\tau^{7}+\frac{117}{2}\,h^{7}\,\tau^{6}+\frac{1515}{16}\,h^{7}\,\tau^{5}+\frac{117}{2}\,h^{7}\,\tau^{4}+\frac{177}{8}\,h^{7}\,\tau^{3}+\right.\\
27\,h^{6}\,\tau^{8}+\frac{2217}{16}\,h^{6}\,\tau^{7}+\frac{22699}{64}\,h^{6}\,\tau^{6}+\frac{14529}{32}\,h^{6}\,\tau^{5}+\frac{22699}{64}\,h^{6}\,\tau^{4}+\frac{2217}{16}\,h^{6}\,\tau^{3}+27\,h^{6}\,\tau^{2}+\frac{3861}{256}\,h^{5}\,\tau^{9}+\\
\frac{4581}{32}\,h^{5}\,\tau^{8}+\frac{145053}{256}\,h^{5}\,\tau^{7}+\frac{38195}{32}\,h^{5}\,\tau^{6}+\frac{393179}{256}\,h^{5}\,\tau^{5}+\frac{38195}{32}\,h^{5}\,\tau^{4}+\frac{145053}{256}\,h^{5}\,\tau^{3}+\frac{4581}{32}\,h^{5}\,\tau^{2}+\\
\frac{3861}{256}\,h^{5}\,\tau+\frac{405}{128}\,h^{4}\,\tau^{10}+\frac{33885}{512}\,h^{4}\,\tau^{9}+\frac{427545}{1024}\,h^{4}\,\tau^{8}+\frac{696081}{512}\,h^{4}\,\tau^{7}+\frac{2728977}{1024}\,h^{4}\,\tau^{6}+\frac{1693935}{512}\,h^{4}\,\tau^{5}
\end{gathered}
$$

$$+\frac{2728977}{1024}h^4\tau^4+\frac{696081}{512}h^4\tau^3+\frac{427545}{1024}h^4\tau^2+\frac{33885}{512}h^4\tau+\frac{405}{128}h^4+\frac{2835}{256}h^3\tau^{10}+\frac{137133}{1024}h^3\tau^9+\frac{715401}{1024}h^3\tau^8+\frac{8427291}{4096}h^3\tau^7+\frac{3901823}{1024}h^3\tau^6+\frac{19088627}{4096}h^3\tau^5+\frac{3901823}{1024}h^3\tau^4+\frac{8427291}{4096}h^3\tau^3+\frac{715401}{1024}h^3\tau^2+\frac{137133}{1024}h^3\tau+\frac{2835}{256}h^3+\frac{6885}{512}h^2\tau^{10}+\frac{292383}{2048}h^2\tau^9+\frac{2777247}{4096}h^2\tau^8+\frac{15475041}{8192}h^2\tau^7+\frac{55720647}{16384}h^2\tau^6+\frac{33757749}{8192}h^2\tau^5+\frac{55720647}{16384}h^2\tau^4+\frac{15475041}{8192}h^2\tau^3+\frac{2777247}{4096}h^2\tau^2+\frac{292383}{2048}h^2\tau+\frac{6885}{512}h^2+\frac{6885}{1024}h\tau^{10}+\frac{149661}{2048}h\tau^9+\frac{1427877}{4096}h\tau^8+\frac{7908723}{8192}h\tau^7+\frac{14134563}{8192}h\tau^6+\frac{8538111}{4096}h\tau^5+\frac{14134563}{8192}h\tau^4+\frac{7908723}{8192}h\tau^3+\frac{1427877}{4096}h\tau^2+\frac{149661}{2048}h\tau+\frac{6885}{1024}h+\frac{1215}{1024}\tau^{10}+\frac{891}{64}\tau^9+\frac{145341}{2048}\tau^8+\frac{423765}{2048}\tau^7+\frac{6259707}{16384}\tau^6+\frac{3823659}{8192}\tau^5+\frac{6259707}{16384}\tau^4+\frac{423765}{2048}\tau^3+\frac{145341}{2048}\tau^2+\frac{891}{64}\tau+\frac{1215}{1024}\Big)$$

```
kacdet_t(4, tau) / kd_rhs;
```

$$37748736$$

```
{num = factorize(num(kd_rhs)), den = den(kd_rhs)};
```

$$\{num = \{\{4h\tau+3\tau^2+6\tau+3,1\},\{4h\tau+6\tau+15,1\},\{h\tau+\tau+2,1\},\{4h+15\tau+6,1\},\{h+2\tau+1,1\},\{4h\tau+2\tau+3,2\},\{4h+3\tau+2,2\},\{h,3\}\},den=16384\tau^5\}$$

```
solve(kd_rhs, h);
```

$$\left\{h=0,h=\tau^{-1}\left(-\frac{1}{2}\tau-\frac{3}{4}\right),h=-\frac{3}{4}\tau-\frac{1}{2},h=\tau^{-1}\left(-\frac{3}{4}\tau^2-\frac{3}{2}\tau-\frac{3}{4}\right),h=\tau^{-1}(-\tau-2),h=-2\tau-1,h=\tau^{-1}\left(-\frac{3}{2}\tau-\frac{15}{4}\right),h=-\frac{15}{4}\tau-\frac{3}{2}\right\}$$

```
% singular vectors
singvec(1, 2, tau);
```

$$l(-1)^2+\tau\, l(-2)$$

```
singvec(2, 1, tau);
```

$$l(-1)^2+\tau^{-1}l(-2)$$

```
singvec(1, 3, tau);
```

$$l(-1)^3+4\tau\,(l(-2)\,l(-1))+2\,l(-3)\,\tau\,(2\tau+1)$$

```
singvec(3, 1, tau);
```

$$l(-1)^3+4\tau^{-1}(l(-2)\,l(-1))+2\tau^{-2}l(-3)\,(\tau+2)$$

```
singvec(1, 4, tau);
```

$$l(-1)^4+9\tau^2l(-2)^2+10\tau\left(l(-2)\,l(-1)^2\right)+2\,l(-3)\,l(-1)\,\tau\,(12\tau+5)+6\,l(-4)\,\tau\,(6\tau^2+4\tau+1)$$

```
singvec(4, 1, tau);
```

$$l(-1)^4+9\tau^{-2}l(-2)^2+10\tau^{-1}\left(l(-2)\,l(-1)^2\right)+2\tau^{-2}l(-3)\,l(-1)\,(5\tau+12)+6\tau^{-3}l(-4)\,(\tau^2+4\tau+6)$$

```
singvec(2, 2, tau);
```

$$l(-1)^4+\tau^{-2}\,l(-2)^2\,\left(\tau^4-2\,\tau^2+1\right)+2\,\tau^{-1}\,l(-2)\,l(-1)^2\,\left(\tau^2+1\right)+2\,\tau^{-1}\,l(-3)\,l(-1)\,\left(\tau^2+3\,\tau+1\right)+ 3\,\tau^{-1}\,l(-4)\,\left(\tau^2+2\,\tau+1\right)$$

```
singvec(1, 5, tau);
```

$$l(-1)^5+64\,\tau^2\,\left(l(-2)^2\,l(-1)\right)+20\,\tau\,\left(l(-2)\,l(-1)^3\right)+6\,l(-3)\,l(-1)^2\,\tau\,(14\,\tau+5)+ 64\,l(-3)\,l(-2)\,\tau^2\,(3\,\tau+1)+12\,l(-4)\,l(-1)\,\tau\,\left(24\,\tau^2+14\,\tau+3\right)+ 8\,l(-5)\,\tau\,\left(72\,\tau^3+66\,\tau^2+23\,\tau+3\right)$$

```
singvec(5, 1, tau);
```

$$l(-1)^5+64\,\tau^{-2}\,\left(l(-2)^2\,l(-1)\right)+20\,\tau^{-1}\,\left(l(-2)\,l(-1)^3\right)+6\,\tau^{-2}\,l(-3)\,l(-1)^2\,(5\,\tau+14)+ 64\,\tau^{-3}\,l(-3)\,l(-2)\,(\tau+3)+12\,\tau^{-3}\,l(-4)\,l(-1)\,\left(3\,\tau^2+14\,\tau+24\right)+ 8\,\tau^{-4}\,l(-5)\,\left(3\,\tau^3+23\,\tau^2+66\,\tau+72\right)$$

```
singvec(1, 6, tau);
```

$$l(-1)^6+225\,\tau^3\,l(-2)^3+259\,\tau^2\,\left(l(-2)^2\,l(-1)^2\right)+35\,\tau\,\left(l(-2)\,l(-1)^4\right)+ 10\,l(-3)^2\,\tau^2\,\left(160\,\tau^2+88\,\tau+13\right)+14\,l(-3)\,l(-1)^3\,\tau\,(16\,\tau+5)+ 2\,l(-3)\,l(-2)\,l(-1)\,\tau^2\,(880\,\tau+259)+6\,l(-4)\,l(-1)^2\,\tau\,\left(216\,\tau^2+112\,\tau+21\right)+ 10\,l(-4)\,l(-2)\,\tau^2\,\left(360\,\tau^2+176\,\tau+31\right)+4\,l(-5)\,l(-1)\,\tau\,\left(1440\,\tau^3+1192\,\tau^2+369\,\tau+42\right)+ 40\,l(-6)\,\tau\,\left(360\,\tau^4+378\,\tau^3+160\,\tau^2+35\,\tau+3\right)$$

```
singvec(6, 1, tau);
```

$$l(-1)^6+225\,\tau^{-3}\,l(-2)^3+259\,\tau^{-2}\,\left(l(-2)^2\,l(-1)^2\right)+35\,\tau^{-1}\,\left(l(-2)\,l(-1)^4\right)+ 10\,\tau^{-4}\,l(-3)^2\,\left(13\,\tau^2+88\,\tau+160\right)+14\,\tau^{-2}\,l(-3)\,l(-1)^3\,(5\,\tau+16)+ 2\,\tau^{-3}\,l(-3)\,l(-2)\,l(-1)\,(259\,\tau+880)+6\,\tau^{-3}\,l(-4)\,l(-1)^2\,\left(21\,\tau^2+112\,\tau+216\right)+ 10\,\tau^{-4}\,l(-4)\,l(-2)\,\left(31\,\tau^2+176\,\tau+360\right)+4\,\tau^{-4}\,l(-5)\,l(-1)\,\left(42\,\tau^3+369\,\tau^2+1192\,\tau+1440\right)+ 40\,\tau^{-5}\,l(-6)\,\left(3\,\tau^4+35\,\tau^3+160\,\tau^2+378\,\tau+360\right)$$

```
singvec(2, 3, tau);
```

$$l(-1)^6+\tau^{-3}\,l(-2)^3\,\left(16\,\tau^4-8\,\tau^2+1\right)+\tau^{-2}\,l(-2)^2\,l(-1)^2\,\left(16\,\tau^4+3\right)+ \tau^{-1}\,l(-2)\,l(-1)^4\,\left(8\,\tau^2+3\right)+2\,\tau^{-2}\,l(-3)^2\,\left(8\,\tau^6+8\,\tau^5-18\,\tau^4-6\,\tau^3+10\,\tau^2+6\,\tau+1\right)+ 2\,\tau^{-1}\,l(-3)\,l(-1)^3\,\left(4\,\tau^3+8\,\tau^2+12\,\tau+3\right)+ 2\,\tau^{-2}\,l(-3)\,l(-2)\,l(-1)\,\left(16\,\tau^5+16\,\tau^4-12\,\tau^3+12\,\tau+3\right)+ 6\,\tau^{-1}\,l(-4)\,l(-1)^2\,\left(4\,\tau^3+18\,\tau^2+12\,\tau+3\right)+ 2\,\tau^{-2}\,l(-4)\,l(-2)\,\left(16\,\tau^5+40\,\tau^4-12\,\tau^3-10\,\tau^2+12\,\tau+5\right)+ 4\,\tau^{-2}\,l(-5)\,l(-1)\,\left(4\,\tau^5+34\,\tau^4+66\,\tau^3+46\,\tau^2+12\,\tau+1\right)+ 4\,\tau^{-2}\,l(-6)\,\left(20\,\tau^5+68\,\tau^4+79\,\tau^3+51\,\tau^2+19\,\tau+3\right)$$

```
singvec(3, 2, tau);
```

$$l(-1)^6+\tau^{-1}\,l(-2)^3\,\left(\tau^4-8\,\tau^2+16\right)+\tau^{-2}\,l(-2)^2\,l(-1)^2\,\left(3\,\tau^4+16\right)+\tau^{-1}\,l(-2)\,l(-1)^4\,\left(3\,\tau^2+8\right) +2\,\tau^{-4}\,l(-3)^2\,\left(\tau^6+6\,\tau^5+10\,\tau^4-6\,\tau^3-18\,\tau^2+8\,\tau+8\right)+2\,\tau^{-2}\,l(-3)\,l(-1)^3\,\left(3\,\tau^3+12\,\tau^2+8\,\tau+4\right)+$$

$$2\tau^{-3}\, l(-3)\, l(-2)\, l(-1)\, \left(3\tau^5+12\tau^4-12\tau^2+16\tau+16\right)+$$
$$6\tau^{-2}\, l(-4)\, l(-1)^2\, \left(3\tau^3+12\tau^2+18\tau+4\right)+$$
$$2\tau^{-3}\, l(-4)\, l(-2)\, \left(5\tau^5+12\tau^4-10\tau^3-12\tau^2+40\tau+16\right)+$$
$$4\tau^{-3}\, l(-5)\, l(-1)\, \left(\tau^5+12\tau^4+46\tau^3+66\tau^2+34\tau+4\right)+$$
$$4\tau^{-3}\, l(-6)\, \left(3\tau^5+19\tau^4+51\tau^3+79\tau^2+68\tau+20\right)$$

```
lincomb(4);
```

$$l(-1)^4+x_3\, l(-2)^2+x_1\, l(-2)\, l(-1)^2+x_2\, l(-3)\, l(-1)+x_4\, l(-4)$$

```
ltoz(lincomb(4), 4);
```

$$x_1\, z_1^2\, z_2+x_2\, z_1\, z_3+x_3\, z_2^2+x_4\, z_4+z_1^4$$

```
allcoeffs(lincomb(4), 4);
```

$$\{x_4, x_3, x_2, x_1, 1\}$$

```
solsingvec(2, 2, tau);
```

$$\left\{\left\{x_1=\frac{2\tau^2+2}{\tau}, x_2=\frac{2\tau^2+6\tau+2}{\tau}, x_3=\frac{\tau^4-2\tau^2+1}{\tau^2}, x_4=\frac{3\tau^2+6\tau+3}{\tau}\right\}\right\}$$

```
singvec(2, 2, tau);
```

$$l(-1)^4+\tau^{-2}\, l(-2)^2\, \left(\tau^4-2\tau^2+1\right)+2\tau^{-1}\, l(-2)\, l(-1)^2\, \left(\tau^2+1\right)+2\tau^{-1}\, l(-3)\, l(-1)\, \left(\tau^2+3\tau+1\right)+$$
$$3\tau^{-1}\, l(-4)\, \left(\tau^2+2\tau+1\right)$$

```
% projection of singvec_{r,s}(tau) by L(-1)->z1, L(-2)->z2, L(-3),L(-4)...->0
proj12(singvec(2, 2, tau));
```

$$\tau^{-2}\, \left(\tau^4\, z_2^2+2\tau^3\, z_1^2\, z_2+\tau^2\, z_1^4-2\tau^2\, z_2^2+2\tau\, z_1^2\, z_2+z_2^2\right)$$

```
proj12_fact(singvec(2, 2, tau));
```

$$\left\{num=\left\{\left\{\tau^2\, z_2+\tau\, z_1^2+2\tau\, z_2+z_2, 1\right\}, \left\{\tau^2\, z_2+\tau\, z_1^2-2\tau\, z_2+z_2, 1\right\}\right\}, den=\tau^2\right\}$$

```
proj12_fact(singvec(1, 1, tau));
```

$$\{num=\{\{z_1, 1\}\}, den=1\}$$

```
proj12_fact(singvec(1, 2, tau));
```

$$\left\{num=\left\{\left\{\tau\, z_2+z_1^2, 1\right\}\right\}, den=1\right\}$$

```
proj12_fact(singvec(1, 3, tau));
```

$$\left\{num=\left\{\left\{4\tau\, z_2+z_1^2, 1\right\}, \{z_1, 1\}\right\}, den=1\right\}$$

```
proj12_fact(singvec(1, 4, tau));
```

$$\left\{num=\left\{\left\{9\tau\, z_2+z_1^2, 1\right\}, \left\{\tau\, z_2+z_1^2, 1\right\}\right\}, den=1\right\}$$

```
proj12_fact(singvec(1, 5, tau));
```

$$\left\{num=\left\{\left\{16\tau\, z_2+z_1^2, 1\right\}, \left\{4\tau\, z_2+z_1^2, 1\right\}, \{z_1, 1\}\right\}, den=1\right\}$$

```
proj12_fact(singvec(1, 6, tau));
```

$$\left\{num=\left\{\left\{25\tau\, z_2+z_1^2, 1\right\}, \left\{9\tau\, z_2+z_1^2, 1\right\}, \left\{\tau\, z_2+z_1^2, 1\right\}\right\}, den=1\right\}$$

```
proj12_fact(singvec(1, 7, tau));
```

$$\left\{ num = \left\{ \left\{ 36\,\tau\,z_2 + z_1^2, 1 \right\}, \left\{ 16\,\tau\,z_2 + z_1^2, 1 \right\}, \left\{ 4\,\tau\,z_2 + z_1^2, 1 \right\}, \left\{ z_1, 1 \right\} \right\}, den = 1 \right\}$$

```
% action of L(m) on w^{h2-h1-h0}
act_ff(l(-3), h0, h0+h1-h2);
```

$$3\,h_0 + h_1 - h_2$$

```
factorize(act_ff(l(-2)*l(-1), h0, h0+h1-h2));
```

$$\left\{ \left\{ 2\,h_0 + h_1 - h_2 + 1, 1 \right\}, \left\{ h_0 + h_1 - h_2, 1 \right\} \right\}$$

```
factorize(act_ff(l(-1)^3, h0, h0+h1-h2));
```

$$\left\{ \left\{ h_0 + h_1 - h_2 + 2, 1 \right\}, \left\{ h_0 + h_1 - h_2 + 1, 1 \right\}, \left\{ h_0 + h_1 - h_2, 1 \right\} \right\}$$

```
% action of singvec_{r,s} on w^{h2-h1-h0}
g_ff(2, 2, cw(1, 1, tau), cw(1, 1, tau) + cw(2, 2, tau) + h2, tau);
```

$$\tau^{-4} \times \left(\frac{9}{256}\,\tau^8 + \frac{3}{16}\,\tau^7\,h_2 + \frac{3}{16}\,\tau^7 - \frac{1}{8}\,\tau^6\,h_2^2 + \frac{1}{2}\,\tau^6\,h_2 + \frac{15}{64}\,\tau^6 - \tau^5\,h_2^3 - \tau^5\,h_2^2 - \frac{3}{16}\,\tau^5\,h_2 - \frac{3}{16}\,\tau^5 + \tau^4\,h_2^4 - \frac{3}{4}\,\tau^4\,h_2^2 - \tau^4\,h_2 - \frac{69}{128}\,\tau^4 - \tau^3\,h_2^3 - \tau^3\,h_2^2 - \frac{3}{16}\,\tau^3\,h_2 - \frac{3}{16}\,\tau^3 - \frac{1}{8}\,\tau^2\,h_2^2 + \frac{1}{2}\,\tau^2\,h_2 + \frac{15}{64}\,\tau^2 + \frac{3}{16}\,\tau\,h_2 + \frac{3}{16}\,\tau + \frac{9}{256} \right)$$

```
% Verify the factorization of g_ff_factor_lhs
solve(g_ff_factor_lhs(r0, s0, r1, s1, h2, i1, j1), h2);
```

$$\Big\{ h_2 = \tau^{-1} \times \left(-\tau^2\,j_1^2 + \tau^2\,j_1\,s_0 + \tau^2\,j_1\,s_1 + \tau^2\,j_1 - \frac{1}{4}\,\tau^2\,s_0^2 - \frac{1}{2}\,\tau^2\,s_0\,s_1 - \frac{1}{2}\,\tau^2\,s_0 - \frac{1}{4}\,\tau^2\,s_1^2 - \frac{1}{2}\,\tau^2\,s_1 - 2\,\tau\,i_1\,j_1 + \tau\,i_1\,s_0 + \tau\,i_1\,s_1 + \tau\,i_1 + \tau\,j_1\,r_0 + \tau\,j_1\,r_1 + \tau\,j_1 - \frac{1}{2}\,\tau\,r_0\,s_0 - \frac{1}{2}\,\tau\,r_0\,s_1 - \frac{1}{2}\,\tau\,r_0 - \frac{1}{2}\,\tau\,r_1\,s_0 - \frac{1}{2}\,\tau\,r_1\,s_1 - \frac{1}{2}\,\tau\,r_1 - \frac{1}{2}\,\tau\,s_0 - \frac{1}{2}\,\tau\,s_1 - i_1^2 + i_1\,r_0 + i_1\,r_1 + i_1 - \frac{1}{4}\,r_0^2 - \frac{1}{2}\,r_0\,r_1 - \frac{1}{2}\,r_0 - \frac{1}{4}\,r_1^2 - \frac{1}{2}\,r_1 \right),$$

$$h_2 = \tau^{-1} \times \left(-\tau^2\,j_1^2 - \tau^2\,j_1\,s_0 + \tau^2\,j_1\,s_1 + \tau^2\,j_1 - \frac{1}{4}\,\tau^2\,s_0^2 + \frac{1}{2}\,\tau^2\,s_0\,s_1 + \frac{1}{2}\,\tau^2\,s_0 - \frac{1}{4}\,\tau^2\,s_1^2 - \frac{1}{2}\,\tau^2\,s_1 - 2\,\tau\,i_1\,j_1 - \tau\,i_1\,s_0 + \tau\,i_1\,s_1 + \tau\,i_1 - \tau\,j_1\,r_0 + \tau\,j_1\,r_1 + \tau\,j_1 - \frac{1}{2}\,\tau\,r_0\,s_0 + \frac{1}{2}\,\tau\,r_0\,s_1 + \frac{1}{2}\,\tau\,r_0 + \frac{1}{2}\,\tau\,r_1\,s_0 - \frac{1}{2}\,\tau\,r_1\,s_1 - \frac{1}{2}\,\tau\,r_1 + \frac{1}{2}\,\tau\,s_0 - \frac{1}{2}\,\tau\,s_1 - i_1^2 - i_1\,r_0 + i_1\,r_1 + i_1 - \frac{1}{4}\,r_0^2 + \frac{1}{2}\,r_0\,r_1 + \frac{1}{2}\,r_0 - \frac{1}{4}\,r_1^2 - \frac{1}{2}\,r_1 \right) \Big\}$$

```
solve(g_ff_factor_rhs(r0, s0, r1, s1, h2, i1, j1), h2);
```

$$\Big\{ h_2 = \tau^{-1} \times \left(-\tau^2\,j_1^2 + \tau^2\,j_1\,s_0 + \tau^2\,j_1\,s_1 + \tau^2\,j_1 - \frac{1}{4}\,\tau^2\,s_0^2 - \frac{1}{2}\,\tau^2\,s_0\,s_1 - \frac{1}{2}\,\tau^2\,s_0 - \frac{1}{4}\,\tau^2\,s_1^2 - \frac{1}{2}\,\tau^2\,s_1 - 2\,\tau\,i_1\,j_1 + \tau\,i_1\,s_0 \right.$$

$$+\tau\, i_1\, s_1 + \tau\, i_1 + \tau\, j_1\, r_0 + \tau\, j_1\, r_1 + \tau\, j_1 - \frac{1}{2}\tau\, r_0\, s_0 - \frac{1}{2}\tau\, r_0\, s_1 - \frac{1}{2}\tau\, r_0 - \frac{1}{2}\tau\, r_1\, s_0 - \frac{1}{2}\tau\, r_1\, s_1 - \frac{1}{2}\tau\, r_1 - \frac{1}{2}\tau\, s_0 - \frac{1}{2}\tau\, s_1 - i_1^2 + i_1\, r_0 + i_1\, r_1 + i_1 - \frac{1}{4} r_0^2 - \frac{1}{2} r_0\, r_1 - \frac{1}{2} r_0 - \frac{1}{4} r_1^2 - \frac{1}{2} r_1\Big),$$

$$h_2 = \tau^{-1}\times \Big(-\tau^2\, j_1^2 - \tau^2\, j_1\, s_0 + \tau^2\, j_1\, s_1 + \tau^2\, j_1 - \frac{1}{4}\tau^2\, s_0^2 + \frac{1}{2}\tau^2\, s_0\, s_1 + \frac{1}{2}\tau^2\, s_0 - \frac{1}{4}\tau^2\, s_1^2 - \frac{1}{2}\tau^2\, s_1 - 2\,\tau\, i_1\, j_1 - \tau\, i_1\, s_0 + \tau\, i_1\, s_1 + \tau\, i_1 - \tau\, j_1\, r_0 + \tau\, j_1\, r_1 + \tau\, j_1 - \frac{1}{2}\tau\, r_0\, s_0 + \frac{1}{2}\tau\, r_0\, s_1 + \frac{1}{2}\tau\, r_0 + \frac{1}{2}\tau\, r_1\, s_0 - \frac{1}{2}\tau\, r_1\, s_1 - \frac{1}{2}\tau\, r_1 + \frac{1}{2}\tau\, s_0 - \frac{1}{2}\tau\, s_1 - i_1^2 - i_1\, r_0 + i_1\, r_1 + i_1 - \frac{1}{4} r_0^2 + \frac{1}{2} r_0\, r_1 + \frac{1}{2} r_0 - \frac{1}{4} r_1^2 - \frac{1}{2} r_1\Big)\Big\}$$

```
g_ff_factor_lhs(r0, s0, r1, s1, h2, i1, j1) / g_ff_factor_rhs(r0, s0, r1, s1, h2, i1, j1);
```

$$1$$

```
% <h2|Phi(w, singvec_{r1,s1}|h1>)|h0> / w^{h2-h1-h0-r1*s1}
f_fusion(1, 1, 2, 2, h2, tau);
```

$$\tau^{-4}\times \Big(\frac{9}{256}\tau^8 - \frac{3}{16}\tau^7\, h_2 + \frac{3}{16}\tau^7 - \frac{1}{8}\tau^6\, h_2^2 - \frac{1}{2}\tau^6\, h_2 + \frac{15}{64}\tau^6 + \tau^5\, h_2^3 - \tau^5\, h_2^2 + \frac{3}{16}\tau^5\, h_2 - \frac{3}{16}\tau^5 + \tau^4\, h_2^4 - \frac{3}{4}\tau^4\, h_2^2 + \tau^4\, h_2 - \frac{69}{128}\tau^4 + \tau^3\, h_2^3 - \tau^3\, h_2^2 + \frac{3}{16}\tau^3\, h_2 - \frac{3}{16}\tau^3 - \frac{1}{8}\tau^2\, h_2^2 - \frac{1}{2}\tau^2\, h_2 + \frac{15}{64}\tau^2 - \frac{3}{16}\tau\, h_2 + \frac{3}{16}\tau + \frac{9}{256}\Big)$$

```
f_fusion_fact(1, 1, 2, 2, h2, tau);
```

$$\{num = \{\{4\,h_2\,\tau + 3\,\tau^2 + 6\,\tau + 3, 1\}, \{4\,h_2\,\tau + 3\,\tau^2 - 2\,\tau - 1, 1\}, \{4\,h_2\,\tau - \tau^2 - 2\,\tau + 3, 1\}, \{4\,h_2\,\tau - \tau^2 - 2\,\tau - 1, 1\}\}, den = 256\,\tau^4\}$$

```
f_fusion_sol(1, 1, 2, 2, h2, tau);
```

$$\left\{h_2 = \frac{\tau^2 + 2\,\tau + 1}{4\,\tau}, h_2 = \frac{\tau^2 + 2\,\tau - 3}{4\,\tau}, h_2 = \frac{-3\,\tau^2 + 2\,\tau + 1}{4\,\tau}, h_2 = \frac{-3\,\tau^2 - 6\,\tau - 3}{4\,\tau}\right\}$$

```
f_rhs := f_fusion_rhs(1, 1, 2, 2, h2, tau);
```

$$f\_rhs := \tau^{-4}\times \Big(\frac{9}{256}\tau^8 - \frac{3}{16}\tau^7\, h_2 + \frac{3}{16}\tau^7 - \frac{1}{8}\tau^6\, h_2^2 - \frac{1}{2}\tau^6\, h_2 + \frac{15}{64}\tau^6 + \tau^5\, h_2^3 - \tau^5\, h_2^2 + \frac{3}{16}\tau^5\, h_2 - \frac{3}{16}\tau^5 + \tau^4\, h_2^4 - \frac{3}{4}\tau^4\, h_2^2 + \tau^4\, h_2 - \frac{69}{128}\tau^4 + \tau^3\, h_2^3 - \tau^3\, h_2^2 + \frac{3}{16}\tau^3\, h_2 - \frac{3}{16}\tau^3 - \frac{1}{8}\tau^2\, h_2^2 - \frac{1}{2}\tau^2\, h_2 + \frac{15}{64}\tau^2 - \frac{3}{16}\tau\, h_2 + \frac{3}{16}\tau + \frac{9}{256}\Big)$$

```
f_fusion(1, 1, 2, 2, h2, tau) / f_rhs;
```

$$1$$

```
{num = factorize(num(f_rhs)), den = den(f_rhs)};
```

$$\{num = \{\{4\,h_2\,\tau + 3\,\tau^2 + 6\,\tau + 3, 1\}, \{4\,h_2\,\tau + 3\,\tau^2 - 2\,\tau - 1, 1\}, \{4\,h_2\,\tau - \tau^2 - 2\,\tau + 3, 1\}, \{4\,h_2\,\tau - \tau^2 - 2\,\tau - 1, 1\}\}, den = 256\,\tau^4\}$$

```
solve(f_rhs, h2);
```

$$\left\{h_2 = \tau^{-1}\left(\frac{1}{4}\tau^2 + \frac{1}{2}\tau + \frac{1}{4}\right), h_2 = \tau^{-1}\left(\frac{1}{4}\tau^2 + \frac{1}{2}\tau - \frac{3}{4}\right), h_2 = \tau^{-1}\left(-\frac{3}{4}\tau^2 + \frac{1}{2}\tau + \frac{1}{4}\right), h_2 = \tau^{-1}\left(-\frac{3}{4}\tau^2 - \frac{3}{2}\tau - \frac{3}{4}\right)\right\}$$

```
% Verify the explicit factorization of f_fusion(r0, s0, r1, s1, h2, tau)

f_fusion(2, 1, 1, 1, h2, tau) / f_fusion_rhs(2, 1, 1, 1, h2, tau);
```

$$1$$

```
f_fusion(2, 2, 1, 2, h2, tau) / f_fusi
on_rhs(2, 2, 1, 2, h2, tau);
```

$$1$$

```
f_fusion(3, 2, 2, 3, h2, tau) / f_fusion_rhs(3, 2, 2, 3, h2, tau);
```

$$1$$

```
f_fusion(2, 1, 2, 4, h2, tau) / f_fusion_rhs(2, 1, 2, 4, h2, tau);
```

$$1$$

```
f_fusion(1, 1, 3, 3, h2, tau) / f_fusion_rhs(1, 1, 3, 3, h2, tau);
```

$$1$$

```
f_fusion(1, 2, 3, 3, h2, tau) / f_fusion_rhs(1, 2, 3, 3, h2, tau);
```

$$1$$

```
f_fusion(2, 2, 4, 2, h2, tau) / f_fusion_rhs(2, 2, 4, 2, h2, tau);
```

$$1$$

```
% the conformal weights of the minimal model for c=c(-p/q)

f_mm45 := f_minimal_model(4, 5);
```

$$f\_mm_{45} := h\left(\frac{7683984}{15625}h^5 - \frac{102773286}{78125}h^4 + \frac{6674020353}{6250000}h^3 - \frac{10040565843}{31250000}h^2 + \frac{3747382947}{125000000}h - \frac{90767061}{125000000}\right)$$

```
factorize(f_mm45);
```

$$\left\{\frac{1}{125000000}, \{480249, 1\}, \{80\,h - 3, 1\}, \{16\,h - 7, 1\}, \{10\,h - 1, 1\}, \{5\,h - 3, 1\}, \{2\,h - 3, 1\}, \{h, 1\}\right\}$$

```
solve(f_mm45, h);
```

$$\left\{h = \frac{7}{16}, h = \frac{3}{80}, h = \frac{3}{5}, h = \frac{3}{2}, h = \frac{1}{10}, h = 0\right\}$$

```
f_minimal_model_fact(2, 3);
```

$$\left\{-\frac{2}{3}, \{h = 0\}\right\}$$

```
f_minimal_model_fact(2, 5);
```

$$\left\{\frac{36}{25}, \left\{h = \frac{-1}{5}, h = 0\right\}\right\}$$

```
f_minimal_model_fact(2, 7);
```

$$\left\{-\frac{1800}{343}, \left\{h = \frac{-2}{7}, h = \frac{-3}{7}, h = 0\right\}\right\}$$

```
f_minimal_model_fact(2, 9);
```

$$\left\{\frac{19600}{729}, \left\{h = \frac{-1}{3}, h = \frac{-2}{3}, h = \frac{-5}{9}, h = 0\right\}\right\}$$

```
f_minimal_model_fact(2, 11);
```

$$\left\{-\frac{28576800}{161051}, \left\{h = \frac{-4}{11}, h = \frac{-7}{11}, h = \frac{-9}{11}, h = \frac{-10}{11}, h = 0\right\}\right\}$$

```
f_minimal_model_fact(3, 4);
```

$$\left\{-\frac{100}{27}, \left\{h = \frac{1}{16}, h = \frac{1}{2}, h = 0\right\}\right\}$$

```
f_minimal_model_fact(3, 5);
```

$$\left\{\frac{802816}{50625}, \left\{h = \frac{3}{4}, h = \frac{1}{5}, h = \frac{-1}{20}, h = 0\right\}\right\}$$

```
f_minimal_model_fact(3, 7);
```

$$\left\{\frac{50751078400}{85766121}, \left\{h = \frac{5}{4}, h = \frac{4}{7}, h = \frac{3}{28}, h = \frac{-1}{7}, h = \frac{-5}{28}, h = 0\right\}\right\}$$

```
f_minimal_model_fact(4, 5);
```

$$\left\{\frac{7683984}{15625}, \left\{h = \frac{7}{16}, h = \frac{3}{80}, h = \frac{3}{5}, h = \frac{3}{2}, h = \frac{1}{10}, h = 0\right\}\right\}$$

```
% Verify explicit simplification of f_minimal_model_factor_lhs(p, q)

f_minimal_model_factor_lhs(p, q, r, s, h);
```

$$-4\,h\,p\,q^{-1}\,s^2 - 4\,h\,p^{-1}\,q\,r^2 + 8\,h\,r\,s + p^2\,q^{-2}\,s^4 - p^2\,q^{-2}\,s^2 - 4\,p\,q^{-1}\,r\,s^3 + 2\,p\,q^{-1}\,r\,s + 2\,p\,q^{-1}\,s^2 - 4\,p^{-1}\,q\,r^3\,s + 2\,p^{-1}\,q\,r^2 + 2\,p^{-1}\,q\,r\,s + p^{-2}\,q^2\,r^4 - p^{-2}\,q^2\,r^2 + 6\,r^2\,s^2 - r^2 - 4\,r\,s - s^2$$

```
f_minimal_model_factor_rhs(p, q, r, w, h);
```

$$-4\,h\,p\,q^{-1}\,s^2 - 4\,h\,p^{-1}\,q\,r^2 + 8\,h\,r\,s + p^2\,q^{-2}\,s^4 - p^2\,q^{-2}\,s^2 - 4\,p\,q^{-1}\,r\,s^3 + 2\,p\,q^{-1}\,r\,s + 2\,p\,q^{-1}\,s^2 - 4\,p^{-1}\,q\,r^3\,s + 2\,p^{-1}\,q\,r^2 + 2\,p^{-1}\,q\,r\,s + p^{-2}\,q^2\,r^4 - p^{-2}\,q^2\,r^2 + 6\,r^2\,s^2 - r^2 - 4\,r\,s - s^2$$

```
f_minimal_model_factor_lhs(p, q, r, s, h) / f_minimal_model_factor_rhs(p, q, r, s, h);
```

$$1$$

```
% table of the conformal weights of the minimal model for c=c(-p/q)

table_minimal_model(2, 3);
```

$$\begin{pmatrix} 0 & 0 \end{pmatrix}$$

```
table_minimal_model(2, 5);
```

$$\begin{pmatrix} 0 & -\frac{1}{5} & -\frac{1}{5} & 0 \end{pmatrix}$$

```
table_minimal_model(2, 7);
```

$$\begin{pmatrix} 0 & -\frac{2}{7} & -\frac{3}{7} & -\frac{3}{7} & -\frac{2}{7} & 0 \end{pmatrix}$$

```
table_minimal_model(2, 9);
```

$$\begin{pmatrix} 0 & -\frac{1}{3} & -\frac{5}{9} & -\frac{2}{3} & -\frac{2}{3} & -\frac{5}{9} & -\frac{1}{3} & 0 \end{pmatrix}$$

```
table_minimal_model(2, 11);
```

$$\begin{pmatrix} 0 & -\frac{4}{11} & -\frac{7}{11} & -\frac{9}{11} & -\frac{10}{11} & -\frac{10}{11} & -\frac{9}{11} & -\frac{7}{11} & -\frac{4}{11} & 0 \end{pmatrix}$$

```
table_minimal_model(2, 13);
```

$$mat\left(\left(0, -\frac{5}{13}, -\frac{9}{13}, -\frac{12}{13}, -\frac{14}{13}, -\frac{15}{13}, -\frac{15}{13}, -\frac{14}{13}, -\frac{12}{13}, -\frac{9}{13}, -\frac{5}{13}, 0\right)\right)$$

```
table_minimal_model(3, 4);
```

$$\begin{pmatrix} 0 & \frac{1}{16} & \frac{1}{2} \\ \frac{1}{2} & \frac{1}{16} & 0 \end{pmatrix}$$

```
table_minimal_model(3, 5);
```

$$\begin{pmatrix} 0 & -\frac{1}{20} & \frac{1}{5} & \frac{3}{4} \\ \frac{3}{4} & \frac{1}{5} & -\frac{1}{20} & 0 \end{pmatrix}$$

```
table_minimal_model(3, 7);
```

$$\begin{pmatrix} 0 & -\frac{5}{28} & -\frac{1}{7} & \frac{3}{28} & \frac{4}{7} & \frac{5}{4} \\ \frac{5}{4} & \frac{4}{7} & \frac{3}{28} & -\frac{1}{7} & -\frac{5}{28} & 0 \end{pmatrix}$$

```
table_minimal_model(3, 8);
```

$$\begin{pmatrix} 0 & -\frac{7}{32} & -\frac{1}{4} & -\frac{3}{32} & \frac{1}{4} & \frac{25}{32} & \frac{3}{2} \\ \frac{3}{2} & \frac{25}{32} & \frac{1}{4} & -\frac{3}{32} & -\frac{1}{4} & -\frac{7}{32} & 0 \end{pmatrix}$$

```
table_minimal_model(3, 10);
```

$$\begin{pmatrix} 0 & -\frac{11}{40} & -\frac{2}{5} & -\frac{3}{8} & -\frac{1}{5} & \frac{1}{8} & \frac{3}{5} & \frac{49}{40} & 2 \\ 2 & \frac{49}{40} & \frac{3}{5} & \frac{1}{8} & -\frac{1}{5} & -\frac{3}{8} & -\frac{2}{5} & -\frac{11}{40} & 0 \end{pmatrix}$$

```
table_minimal_model(3, 11);
```

$$\begin{pmatrix} 0 & -\frac{13}{44} & -\frac{5}{11} & -\frac{21}{44} & -\frac{4}{11} & -\frac{5}{44} & \frac{3}{11} & \frac{35}{44} & \frac{16}{11} & \frac{9}{4} \\ \frac{9}{4} & \frac{16}{11} & \frac{35}{44} & \frac{3}{11} & -\frac{5}{44} & -\frac{4}{11} & -\frac{21}{44} & -\frac{5}{11} & -\frac{13}{44} & 0 \end{pmatrix}$$

```
table_minimal_model(3, 13);
```

$$mat\left(\left(0,-\frac{17}{52},-\frac{7}{13},-\frac{33}{52},-\frac{8}{13},-\frac{25}{52},-\frac{3}{13},\frac{7}{52},\frac{8}{13},\frac{63}{52},\frac{25}{13},\frac{11}{4}\right),\right.$$
$$\left.\left(\frac{11}{4},\frac{25}{13},\frac{63}{52},\frac{8}{13},\frac{7}{52},-\frac{3}{13},-\frac{25}{52},-\frac{8}{13},-\frac{33}{52},-\frac{7}{13},-\frac{17}{52},0\right)\right)$$

```
table_minimal_model(4, 5);
```

$$\begin{pmatrix} 0 & \frac{1}{10} & \frac{3}{5} & \frac{3}{2} \\ \frac{7}{16} & \frac{3}{80} & \frac{3}{80} & \frac{7}{16} \\ \frac{3}{2} & \frac{3}{5} & \frac{1}{10} & 0 \end{pmatrix}$$

```
table_minimal_model(4, 7);
```

$$\begin{pmatrix} 0 & -\frac{1}{14} & \frac{1}{7} & \frac{9}{14} & \frac{10}{7} & \frac{5}{2} \\ \frac{13}{16} & \frac{27}{112} & -\frac{5}{112} & -\frac{5}{112} & \frac{27}{112} & \frac{13}{16} \\ \frac{5}{2} & \frac{10}{7} & \frac{9}{14} & \frac{1}{7} & -\frac{1}{14} & 0 \end{pmatrix}$$

```
table_minimal_model(4, 9);
```

$$\begin{pmatrix} 0 & -\frac{1}{6} & -\frac{1}{9} & \frac{1}{6} & \frac{2}{3} & \frac{25}{18} & \frac{7}{3} & \frac{7}{2} \\ \frac{19}{16} & \frac{25}{48} & \frac{11}{144} & -\frac{7}{48} & -\frac{7}{48} & \frac{11}{144} & \frac{25}{48} & \frac{19}{16} \\ \frac{7}{2} & \frac{7}{3} & \frac{25}{18} & \frac{2}{3} & \frac{1}{6} & -\frac{1}{9} & -\frac{1}{6} & 0 \end{pmatrix}$$

```
table_minimal_model(4, 11);
```

$$mat\left(\left(0,-\frac{5}{22},-\frac{3}{11},-\frac{3}{22},\frac{2}{11},\frac{15}{22},\frac{15}{11},\frac{49}{22},\frac{36}{11},\frac{9}{2}\right),\right.$$
$$\left.\left(\frac{25}{16},\frac{147}{176},\frac{51}{176},-\frac{13}{176},-\frac{45}{176},-\frac{45}{176},-\frac{13}{176},\frac{51}{176},\frac{147}{176},\frac{25}{16}\right),\left(\frac{9}{2},\frac{36}{11},\frac{49}{22},\frac{15}{11},\frac{15}{22},\frac{2}{11},-\frac{3}{22},-\frac{3}{11},-\frac{5}{22},0\right)\right)$$

```
table_minimal_model(4, 13);
```

$$mat\left(\left(0,-\frac{7}{26},-\frac{5}{13},-\frac{9}{26},-\frac{2}{13},\frac{5}{26},\frac{9}{13},\frac{35}{26},\frac{28}{13},\frac{81}{26},\frac{55}{13},\frac{11}{2}\right),\right.$$
$$\left(\frac{31}{16},\frac{243}{208},\frac{115}{208},\frac{19}{208},-\frac{45}{208},-\frac{77}{208},-\frac{77}{208},-\frac{45}{208},\frac{19}{208},\frac{115}{208},\frac{243}{208},\frac{31}{16}\right),$$
$$\left.\left(\frac{11}{2},\frac{55}{13},\frac{81}{26},\frac{28}{13},\frac{35}{26},\frac{9}{13},\frac{5}{26},-\frac{2}{13},-\frac{9}{26},-\frac{5}{13},-\frac{7}{26},0\right)\right)$$

```
table_minimal_model(5, 6);
```

$$\begin{pmatrix} 0 & \frac{1}{8} & \frac{2}{3} & \frac{13}{8} & 3 \\ \frac{2}{5} & \frac{1}{40} & \frac{1}{15} & \frac{21}{40} & \frac{7}{5} \\ \frac{7}{5} & \frac{21}{40} & \frac{1}{15} & \frac{1}{40} & \frac{2}{5} \\ 3 & \frac{13}{8} & \frac{2}{3} & \frac{1}{8} & 0 \end{pmatrix}$$

```
table_minimal_model(5, 7);
```

$$\begin{pmatrix} 0 & \frac{1}{28} & \frac{3}{7} & \frac{33}{28} & \frac{16}{7} & \frac{15}{4} \\ \frac{11}{20} & \frac{3}{35} & -\frac{3}{140} & \frac{8}{35} & \frac{117}{140} & \frac{9}{5} \\ \frac{9}{5} & \frac{117}{140} & \frac{8}{35} & -\frac{3}{140} & \frac{3}{35} & \frac{11}{20} \\ \frac{15}{4} & \frac{16}{7} & \frac{33}{28} & \frac{3}{7} & \frac{1}{28} & 0 \end{pmatrix}$$

```
table_minimal_model(5, 8);
```

$$\begin{pmatrix} 0 & -\frac{1}{32} & \frac{1}{4} & \frac{27}{32} & \frac{7}{4} & \frac{95}{32} & \frac{9}{2} \\ \frac{7}{10} & \frac{27}{160} & -\frac{1}{20} & \frac{7}{160} & \frac{9}{20} & \frac{187}{160} & \frac{11}{5} \\ \frac{11}{5} & \frac{187}{160} & \frac{9}{20} & \frac{7}{160} & -\frac{1}{20} & \frac{27}{160} & \frac{7}{10} \\ \frac{9}{2} & \frac{95}{32} & \frac{7}{4} & \frac{27}{32} & \frac{1}{4} & -\frac{1}{32} & 0 \end{pmatrix}$$

```
table_minimal_model(5, 9);
```

$$\begin{pmatrix} 0 & -\frac{1}{12} & \frac{1}{9} & \frac{7}{12} & \frac{4}{3} & \frac{85}{36} & \frac{11}{3} & \frac{21}{4} \\ \frac{17}{20} & \frac{4}{15} & -\frac{7}{180} & -\frac{1}{15} & \frac{11}{60} & \frac{32}{45} & \frac{91}{60} & \frac{13}{5} \\ \frac{13}{5} & \frac{91}{60} & \frac{32}{45} & \frac{11}{60} & -\frac{1}{15} & -\frac{7}{180} & \frac{4}{15} & \frac{17}{20} \\ \frac{21}{4} & \frac{11}{3} & \frac{85}{36} & \frac{4}{3} & \frac{7}{12} & \frac{1}{9} & -\frac{1}{12} & 0 \end{pmatrix}$$

```
table_minimal_model(5, 11);
```

$$mat\left(\left(0, -\frac{7}{44}, -\frac{1}{11}, \frac{9}{44}, \frac{8}{11}, \frac{65}{44}, \frac{27}{11}, \frac{161}{44}, \frac{56}{11}, \frac{27}{4}\right), \left(\frac{23}{20}, \frac{27}{55}, \frac{13}{220}, -\frac{8}{55}, -\frac{27}{220}, \frac{7}{55}, \frac{133}{220}, \frac{72}{55}, \frac{493}{220}, \frac{17}{5}\right),\right.$$
$$\left.\left(\frac{17}{5}, \frac{493}{220}, \frac{72}{55}, \frac{133}{220}, \frac{7}{55}, -\frac{27}{220}, -\frac{8}{55}, \frac{13}{220}, \frac{27}{55}, \frac{23}{20}\right), \left(\frac{27}{4}, \frac{56}{11}, \frac{161}{44}, \frac{27}{11}, \frac{65}{44}, \frac{8}{11}, \frac{9}{44}, -\frac{1}{11}, -\frac{7}{44}, 0\right)\right)$$

```
table_minimal_model(5, 12);
```

$$mat\left(\left(0, -\frac{3}{16}, -\frac{1}{6}, \frac{1}{16}, \frac{1}{2}, \frac{55}{48}, 2, \frac{49}{16}, \frac{13}{3}, \frac{93}{16}, \frac{15}{2}\right), \left(\frac{13}{10}, \frac{49}{80}, \frac{2}{15}, -\frac{11}{80}, -\frac{1}{5}, -\frac{13}{240}, \frac{3}{10}, \frac{69}{80}, \frac{49}{30}, \frac{209}{80}, \frac{19}{5}\right),\right.$$
$$\left.\left(\frac{19}{5}, \frac{209}{80}, \frac{49}{30}, \frac{69}{80}, \frac{3}{10}, -\frac{13}{240}, -\frac{1}{5}, -\frac{11}{80}, \frac{2}{15}, \frac{49}{80}, \frac{13}{10}\right), \left(\frac{15}{2}, \frac{93}{16}, \frac{13}{3}, \frac{49}{16}, 2, \frac{55}{48}, \frac{1}{2}, \frac{1}{16}, -\frac{1}{6}, -\frac{3}{16}, 0\right)\right)$$

```
table_minimal_model(5, 13);
```

$$mat\left(\left(0, -\frac{11}{52}, -\frac{3}{13}, -\frac{3}{52}, \frac{4}{13}, \frac{45}{52}, \frac{21}{13}, \frac{133}{52}, \frac{48}{13}, \frac{261}{52}, \frac{85}{13}, \frac{33}{4}\right),\right.$$
$$\left(\frac{29}{20}, \frac{48}{65}, \frac{57}{260}, -\frac{7}{65}, -\frac{63}{260}, -\frac{12}{65}, \frac{17}{260}, \frac{33}{65}, \frac{297}{260}, \frac{128}{65}, \frac{777}{260}, \frac{21}{5}\right),$$
$$\left(\frac{21}{5}, \frac{777}{260}, \frac{128}{65}, \frac{297}{260}, \frac{33}{65}, \frac{17}{260}, -\frac{12}{65}, -\frac{63}{260}, -\frac{7}{65}, \frac{57}{260}, \frac{48}{65}, \frac{29}{20}\right),$$
$$\left.\left(\frac{33}{4}, \frac{85}{13}, \frac{261}{52}, \frac{48}{13}, \frac{133}{52}, \frac{21}{13}, \frac{45}{52}, \frac{4}{13}, -\frac{3}{52}, -\frac{3}{13}, -\frac{11}{52}, 0\right)\right)$$

```
;
```

```
end;
```

3: